

This copy of the Lingua Franca handbook for the c target was created on Saturday, April 23, 2022 against commit fdb373.

Table of Contents

<u>A First Reactor</u> Writing your first Lingua Franca reactor.

<u>Inputs and Outputs</u> Inputs, outputs, and reactions in Lingua Franca.

<u>Parameters and State</u>

Parameters and state variables in Lingua Franca.

<u>Variables</u>

<u>Time and Timers</u>

<u>Composing Reactors</u>

<u>Reactions and Methods</u>

Time and timers in Lingua Franca.

Composing reactors in Lingua Franca.

Reactions and methods in Lingua Franca.

<u>Causality Loops</u>

Causality loops in Lingua Franca.

<u>Extending Reactors</u>

Extending reactors in Lingua Franca.

<u>Actions</u> Actions in Lingua Franca.

<u>Superdense Time</u> Superdense time in Lingua Franca.

<u>Deadlines</u> Deadlines in Lingua Franca.

<u>Multiports and Banks</u> Multiports and Banks of Reactors.

<u>Preambles and Methods</u> Defining functions and methods in Lingua Franca.

<u>Distributed Execution</u> Distributed Execution (preliminary)

<u>Termination</u> Terminating a Lingua Franca execution.

A First Reactor

See the <u>requirements</u> for using this target.

Minimal Example

A minimal but complete Lingua Franca file with one reactor is this:

```
target C;
main reactor {
    reaction(startup) {=
        printf("Hello World.\n");
    =}
}
```

Every Lingua Franca program begins with a <u>target specification</u> that specifies the language in which reactions are written. This is also the language of the program(s) generated by the Lingua Franca code generator.

Every LF program also has a **main** reactor, which is the top level of a hierarchy of contained and interconnected reactors. The above simple example has no contained reactors.

The **main** reactor above has a single **reaction**, which is triggered by the **startup** trigger. This trigger causes the reaction to execute at the start of the program. The body of the reaction, delimitted by $\{= \ldots =\}$, is ordinary C code which, as we will see, has access to a number of functions and variables specific to Lingua Franca.

Examples

Examples of Lingua Franca programs can be found in the examples-lingua-franca repository.

The <u>regression tests</u> have a rich set of examples that illustrate every feature of the language.

Structure of an LF Project

The Lingua Franca tools assume that LF programs are put into a file with a .lf extension that is stored somewhere within a directory called src. To compile and run the above example, choose a **project root** directory, create a src directory within that, and put the above code into a file called,

say, src/HelloWorld.lf. You can compile the code on the <u>command line</u>, within <u>Visual Studio</u> Code, or within the Epoch IDE. On the command line this will look like this:

```
> lfc src/Minimal.lf
... output from the code generator and compiler ...
```

Reactor Block

A **reactor** is a software component that reacts to input events, timer events, and internal events. It has private state variables that are not visible to any other reactor. Its reactions can consist of altering its own state, sending messages to other reactors, or affecting the environment through some kind of actuation or side effect (e.g., printing a message, as in the above HelloWorld example).

The general structure of a reactor definition is as follows:

Contents within square brackets are optional, contents within <...> are user-defined, and each line may appear zero or more times, as explained in the next pages. Parameters, inputs, outputs, timers, actions, and contained reactors all have names, and the names are required to be distinct from one another.

If the **reactor** keyword is preceded by **main**, then this reactor will be instantiated and run by the generated code.

Any number of reactors may be defined in one file, and a **main** reactor need not be given a name, but if it is given a name, then that name must match the file name.

Reactors may extend other reactors, inheriting their properties, and a file may import reactors from other files. If an imported LF file contains a **main** reactor, that reactor is ignored (it will not be imported). This makes it easy to create a library of reusable reactors that each come with a test case or demonstration in the form of a main reactor.

Comments

Lingua Franca files can have C/C++/Java-style comments and/or Python-style comments. All of the following are valid comments:

Inputs and Outputs

In this section, we will endow reactors with inputs and outputs.

Input and Output Declarations

Input and output declarations have the form:

For example, the following reactor doubles its input and sends the result to the output:

```
target C;
reactor Double {
    input x:int;
    output y:int;
    reaction(x) -> y {=
        SET(y, x->value * 2);
    =}
}
```

Notice how the input value is accessed and how the output value is set. This is done differently for each target language. See the <u>Target Language Reference</u> for detailed documentation of these mechanisms. Setting an output within a reaction will trigger downstream reactions at the same <u>Logical Time</u> that the reaction is invoked (or, more precisely, at the same <u>tag</u>). If a particular output port is set more than once at any tag, the last set value will be the one that downstream reactions see. Since the order in which reactions of a reactor are invoked at a logical time is deterministic, and whether inputs are present depends only on their timestamps, the final value set for an output will also be deterministic.

The **reaction** declaration above indicates that an input event on port x is a **trigger** and that an output event on port y is a (potential) **effect**. A reaction can declare more than one trigger or effect by just listing them separated by commas. For example, the following reactor has two triggers and tests each input for presence before using it:

```
target C;
reactor Destination {
   input x:int;
   input y:int;
   reaction(x, y) {=
      int sum = 0;
      if (x->is_present) {
        sum += x->value;
      }
      if (y->is_present) {
        sum += y->value;
      }
      printf("Received %d.\n", sum);
      =}
}
```

NOTE: if a reaction fails to test for the presence of an input and reads its value anyway, then the result it will get is target dependent. In the C target, the value read will be the most recently seen input value, or, if no input event has occurred at an earlier logical time, then zero or NULL, depending on the datatype of the input.

Triggers, Effects, and Uses

The general form of a reaction is

The **triggers** field can be a comma-separated list of input ports, <u>output ports of contained reactors</u>, <u>timers, actions</u>, or the special events **startup** and **shutdown**. There must be at least one trigger for each reaction. A reaction with a **startup** trigger is invoked when the program begins executing, and a reaction with a **shutdown** trigger is invoked at the end of execution.

The **uses** field, which is optional, specifies input ports (or <u>output ports of contained reactors</u>) that do not trigger execution of the reaction but may be read by the reaction.

The **effects** field, which is also optional, is a comma-separated lists of output ports ports, <u>input</u> <u>ports of contained reactors</u>, or <u>actions</u>.

Mutable Inputs

Normally, a reaction does not modify the value of an input. An input is said to be **immutable**. The degree to which this is enforced varies by target language. Most of the target languages make it rather difficult to enforce, so the programmer needs to avoid modifying the input. Modifying an input value may lead to nondeterministic results.

Occassionally, it is useful to modify an input. For example, the input may be a large data structure, and a reaction may wish to make a small modification and forward the result to an output. To accomplish this, the programmer should declare the input **mutable** as follows:

mutable input <name>;

This is a directive to the code generator indicating that reactions that read this input may also modify the value of the input. The code generator will attempt to optimize the scheduling to avoid copying the input value, but this may not be possible, in which case it will automatically insert a copy operation, making it safe to modify the input. The target-spefic reference documentation has more details about how this works.

Parameters and State Variables

Parameter Declaration

A reactor class definition can parameterized as follows:

Types ending with a * are treated specially by the C target. See the <u>Target Language Reference</u>.

To use strings conveniently in the C target, the "type" string is an alias for {=const char*=}.

Each parameter must have a *default value*, written (<expr>). An expression may be a numeric contant, a string enclosed in quotation marks, a time value such as 10 msec, a list of values, or target-language code enclosed in {= ... =}, for example. See Expressions for full details on what expressions are valid.

For example, the **Double** reactor on the <u>previous page</u> can be replaced with a more general parameterized reactor **Scale** as follows:

```
target C;
reactor Scale(factor:int(2)) {
    input x:int;
    output y:int;
    reaction(x) -> y {=
        SET(y, x->value * self->factor);
    =}
}
```

This reactor, given any input event x will produce an output y with value equal to the input scaled by the factor parameter. The default value of the factor parameter is 2, but this can be changed when the Scale reactor is instantiated.

Notice how, within the body of a reaction, the code accesses the parameter value. This is different for each target language. In the C target, a self struct is provided that contains the parameter values.

State Declaration

A reactor declares a state variable as follows:

The <value> is an initial value and, like parameter values, can be given as an expression or target language code with delimiters {= ... =} . The initial value can also be given as a parameter name. The value can be accessed and modified in a target-language-dependent way as illustrated by the following example:

```
target C;
reactor Count {
    state count:int(0);
    output y:int;
    timer t(0, 100 msec);
    reaction(t) -> y {=
        SET(y, self->count++);
    =}
}
```

This reactor has an integer state variable named **count**, and each time its reaction is invoked, it outputs the value of that state variable and increments it. The reaction is trigger by a **timer**, discussed in the next section.

Time and Timers

Logical Time

A key property of Lingua Franca is **logical time**. All events occur at an instant in logical time. By default, the runtime system does its best to align logical time with **physical time**, which is some measurement of time on the execution platform. The **lag** is defined to be physical time minus logical time, and the goal of the runtime system is maintain a small non-negative lag.

The **lag** is allowed to go negative only if the <u>fast target property</u> or the <u>--fast</u> is set to <u>true</u>. In that case, the program will execute as fast as possible with no regard to physical time.

Time Values

A time value is given with units (unless the value is 0, in which case the units can be omitted). The allowable units are:

- For nanoseconds: ns , nsec , or nsecs
- For microseconds: us , usec , or usecs
- For milliseconds: ms , msec , or msecs
- For seconds: s, sec, secs, second, or seconds
- For minutes: min, minute, mins, or minutes
- For hours: h, hour, or hours
- For days: d , day , or days
- For weeks: week or weeks

The following example illustrates using time values for parameters and state variables:

```
target C;
main reactor SlowingClock(start:time(100 msec), incr:time(100 msec)) {
    state interval:time(start);
    logical action a;
    reaction(startup) -> a {=
        schedule(a, self->start);
    =}
    reaction(a) -> a {=
        instant t elapsed logical time = get elapsed logical time();
        printf("Logical time since start: \%lld nsec.\n",
            elapsed logical time
        );
        self->interval += self->incr;
        schedule(a, self->interval);
    =}
}
```

This has two time parameters, start and incr, each with default value 100 msec. This parameter is used to initialize the interval state variable, which also stores a time. The logical action a, explained below, is used to schedule events to occur at time start after program startup and then at intervals that are increased each time by incr. The result of executing this program will look like this:

```
Logical time since start: 100000000 nsec.
Logical time since start: 300000000 nsec.
Logical time since start: 600000000 nsec.
Logical time since start: 1000000000 nsec.
...
```

Timers

The simplest use of logical time in Lingua Franca is to invoke a reaction periodically. This is done by first declaring a **timer** using this syntax:

```
timer <name>(<offset>, <period>);
```

The <period> , which is optional, specifies the time interval between timer events. The <offset> , which is also optional, specifies the (logical) time interval between when the program starts executing and the first timer event. If no period is given, then the timer event occurs only once. If neither an offset nor a period is specified, then one timer event occurs at program start, simultaneous with the startup event.

The period and offset are given by a number and a units, for example, 10 msec. See the expressions documentation for allowable units. Consider the following example:

```
target C;
main reactor Timer {
    timer t(0, 1 sec);
    reaction(t) {=
        printf("Logical time is %lld.\n", get_logical_time());
        =}
}
```

This specifies a timer named t that will first trigger at the start of execution and then repeatedly trigger at intervals of one second. Notice that the time units can be left off if the value is zero.

Each target provides a built-in function for retrieving the logical time at which the reaction is invoked, get_logical_time(). On most platforms (with the exception of some embedded platforms), the returned value is a 64-bit number representing the number of nanoseconds that have elapsed since January 1, 1970. Executing the above displays something like the following:

```
Logical time is 1648402121312985000.
Logical time is 1648402122312985000.
Logical time is 1648402123312985000.
```

The output lines appear at one second intervals unless the fast option has been specified.

Elapsed Time

The times above are a bit hard to read, so, for convenience, each target provides a built-in function to retrieve the *elapsed* time. For example:

See the <u>Target Language Reference</u> for the full set of functions provided for accessing time values.

Executing this program will produce something like this:

```
Elapsed logical time is 0.
Elapsed logical time is 1000000000.
Elapsed logical time is 2000000000.
```

Comparing Logical and Physical Times

The following program compares logical and physical times:

Execution will show something like this:

```
Elapsed logical time: 0, physical time: 855000, lag: 855000
Elapsed logical time: 1000000000, physical time: 1004714000, lag: 4714000
Elapsed logical time: 2000000000, physical time: 2004663000, lag: 4663000
Elapsed logical time: 3000000000, physical time: 3000210000, lag: 210000
```

In this case, the lag varies from a few hundred microseconds to a small number of milliseconds. The amount of lag will depend on the execution platform.

Simultaneity and Instantaneity

If two timers have the same *offset* and *period*, then their events are logically simultaneous. No observer will be able to see that one timer has triggered and the other has not.

A reaction is always invoked at a well-defined logical time, and logical time does not advance during its execution. Any output produced by the reaction will be **logically simultaneous** with the

input. In other words, reactions are **logically instantaneous** (for an exception, see <u>Logical Execution Time</u>). Physical time, however, does elapse during execution of a reaction.

Timeout

By default, a Lingua Franca program will terminate when there are no more events to process. If there is a timer with a non-zero period, then there will always be more events to process, so the default execution will be unbounded. To specify a finite execution horizon, you can either specify a timeout target property or a [--timeout command-line option] (ocs/handbook/target-declaration#command-line-arguments). For example, the following timeout property will cause the above timer with a period of one second to terminate after 11 events:

```
target C {
    timeout: 10 sec
}
```

Startup and Shutdown

To cause a reaction to be invoked at the start of execution, a special **startup** trigger is provided:

The **startup** trigger is equivalent to a timer with no *offset* or *period*.

To cause a reaction to be invoked at the end of execution, a special **shutdown** trigger is provided. Consider the following reactor, commonly used to build regression tests:

```
target C;
reactor TestCount(start:int(0), stride:int(1), num inputs:int(1)) {
    state count:int(start);
    state inputs received:int(0);
    input x:int;
    reaction(x) {=
        printf("Received %d.\n", x->value);
        if (x->value != self->count) {
            printf("ERROR: Expected %d.\n", self->count);
            exit(1);
        }
        self->count += self->stride;
        self->inputs received++;
    =}
    reaction(shutdown) {=
        printf("Shutdown invoked.\n");
        if (self->inputs received != self->num inputs) {
            printf("ERROR: Expected to receive %d inputs, but got %d.\n",
                self->num inputs,
                self->inputs received
            );
            exit(2);
        }
    =}
}
```

This reactor tests its inputs against expected values, which are expected to start with the value given by the start parameter and increase by stride with each successive input. It expects to receive a total of num_inputs input events. It checks the total number of inputs received in its shutdown reaction.

The **shutdown** trigger typically occurs at <u>microstep</u> 0, but may occur at a larger microstep. See <u>Superdense Time</u> and <u>Termination</u>.

Composing Reactors

Contained Reactors

Reactors can contain instances of other reactors defined in the same file or in an imported file.

Assume the Count and Scale reactors defined in <u>Parameters and State Variables</u> are stored in files Count.lf and Scale.lf, respectively, and that the <u>TestCount</u> reactor from <u>Time and</u> <u>Timers</u> is stored in <u>TestCount.lf</u>. Then the following program composes one instance of each of the three:

```
target C {
    timeout: 1 sec,
    fast: true
}
import Count from "Count.lf";
import Scale from "Scale.lf";
import TestCount from "TestCount.lf";

main reactor RegressionTest {
    c = new Count();
    s = new Scale(factor = 4);
    t = new TestCount(stride = 4, num_inputs = 11);
    c.y -> s.x;
    s.y -> t.x;
}
```

Diagrams

As soon as programs consist of more than one reactor, it becomes particularly useful to reference the diagrams that are automatically created and displayed by the Lingua Franca IDEs. The diagram for the above program is as follows:



In this diagram, the timer is represented by a clock-like icon, the reactions by chevron shapes, and the **shutdown** event by a diamond. If there were a **startup** event in this program, it would appear as a circle.

Creating Reactor Instances

An instance is created with the syntax:

```
<instance name> = new <class name>(<parameters>)
```

A bank with several instances can be created in one such statement, as explained in the <u>banks of</u> reactors documentation.

The <parameters> argument is a comma-separated list of assignments:

```
<parameter_name> = <value>, ...
```

Like the default value for parameters, <value> can be a numeric contant, a string enclosed in quotation marks, a time value such as 10 msec, target-language code enclosed in {= ... =}, or any of the list forms described in Expressions.

Connections

Connections between ports are specified with the syntax:

```
<source_port_reference> -> <destination_port_reference>
```

where the port references are either <instance_name>.<port_name> or just <port_name>, where the latter form is used for connections that cross hierarchical boundaries, as illustrated in the next section.

On the left and right of a connection statement, you can put a comma-separated list. For example, the above pair of connections can be written,

```
c.y, s.y \rightarrow s.x, t.x
```

The only constraint is that the total number of channels on the left match the total number on the right.

A destination port (on the right) can only be connected to a single source port (on the left). However, a source port may be connected to multiple destinations, as in the following example:

```
reactor A {
    output y:int
}
reactor B {
    input x:int
}
main reactor {
    a = new A()
    b1 = new B()
    b2 = new B()
    a.y -> b1.x
    a.y -> b2.x
}
```



Lingua Franca provides a convenient shortcut for such multicast connections, where the above two lines can be replaced by one as follows:

$$(a.y) + -> b1.x, b2.x$$

The enclosing (...)+ means to repeat the enclosed comma-separated list of sources however many times is needed to provide inputs to all the sinks on the right of the connection ->.

Import Statement

An import statement has the form:

```
import <reactor class> as <alias2> from "<path>"
```

where <reactor class> and <alias> can be a comma-separated list to import multiple reactors from the same file. The <path> specifies another .lf file relative to the location of the current file. The as <alias> portion is optional and specifies alternative class names to use in the new statements.

Hierarchy

Reactors can be composed in arbitrarily deep hierarchies. For example, the following program combines the Count and Scale reactors within on Container:

```
target C;
import Count from "Count.lf";
import Scale from "Scale.lf";
import TestCount from "TestCount.lf";
reactor Container(stride:int(2)) {
    output y:int;
    c = new Count();
    s = new Scale(factor = stride);
    c.y -> s.x;
    s.y -> y;
}
main reactor Hierarchy {
    c = new Container(stride = 4);
    t = new TestCount(stride = 4, num_inputs = 11);
    c.y -> t.x;
}
Lingua Franca diagram
```

The Container has a parameter named stride, whose value is passed to the factor parameter of the Scale reactor. The line

```
s.y \rightarrow y;
```

establishes a connection across levels of the hierarchy. This propagates the output of a contained reactor to the output of the container. A similar notation may be used to propagate the input of a container to the input of a contained reactor,

```
X \rightarrow S.X;
```

Connections with Logical Delays

Connections may include a **logical delay** using the **after** keyword, as follows:

```
<source_port_reference> -> <destination_port_reference> after <time_val
where <time_value> can be any of the forms described in <a href="Expressions">Expressions</a>.
```

The **after** keyword specifies that the logical time of the event delivered to the destination port will be larger than the logical time of the reaction that wrote to source port. The time value is required to be non-negative, but it can be zero, in which case the input event at the receiving end will be one <u>microstep</u> later than the event that triggered it.

Physical Connections

A subtle and rarely used variant of the -> connection is a **physical connection**, denoted ~> . For example:

```
main reactor {
    a = new A();
    b = new B();
    a.y ~> b.x;
}
```

This is rendered in by the diagram synthesizer as follows:



In such a connection, the logical time at the recipient is derived from the local physical clock rather than being equal to the logical time at the sender. The physical time will always exceed the logical time of the sender (unless fast is set to true), so this type of connection incurs a nondeterministic positive logical time delay. Physical connections are useful sometimes in Distributed-Execution in situations where the nondeterministic logical delay is tolerable. Such connections are more efficient because timestamps need not be transmitted and messages do not need to flow through through a centralized coordinator (if a centralized coordinator is being used).

Reactions and Methods

Reaction Order

A reactor may have multiple reactions, and more than one reaction may be enabled at any given tag. In Lingua Franca semantics, if two or more reactions of the same reactor are **simultaneously enabled**, then they will be invoked sequentially in the order in which they are declared. More strongly, the reactions of a reactor are **mutually exclusive** and are invoked in tag order primarily and declaration order secondarily. Consider the following example:

```
target C {
    timeout: 3 secs
}
main reactor Alignment {
    state s:int(0);
    timer t1(100 msec, 100 msec);
    timer t2(200 msec, 200 msec);
    timer t4(400 msec, 400 msec);
    reaction(t1) {=
        self->s += 1;
    =}
    reaction(t2) {=
        self->s -= 2;
    =}
    reaction(t4) {=
        printf("s = %d\n", self->s);
    =}
}
```

Every 100 ms, this increments the state variable s by 1, every 200 ms, it decrements s by 2, and every 400 ms, it prints the value of s. When these reactions align, they are invoked in declaration order, and, as a result, the printed value of s is always 0.

Overwriting Outputs

Just as the reactions of the Alignment reactor overwrite the state variable s, logically simultaneous reactions can overwrite outputs. Consider the following example:

```
target C;
reactor Overwriting {
    output y:int;
    state s:int(0);
    timer t1(100 msec, 100 msec);
    timer t2(200 msec, 200 msec);
    reaction(t1) -> y {=
        self->s += 1;
        SET(y, self->s);
    =}
    reaction(t2) -> y {=
        self->s -= 2;
        SET(y, self->s);
    =}
}
```

Here, the reaction to t1 will set the output to 1 or 2, but every time it sets it to 2, the second reaction (to t2) will overwrite the output with the value 0. As a consequence, the outputs will be 1, 0, 1, 0, ... deterministically.

Reacting to Outputs of Contained Reactors

A reaction may be triggered by the an input to the reactor, but also by an output of a contained reactor, as illustrated in the following example:

```
target C;
import Overwriting from "Overwriting.lf";
main reactor {
    s = new Overwriting();
    reaction(s.y) {=
        if (s.y->value != 0 && s.y->value != 1) {
            error_print_and_exit("Outputs should only be 0 or 1!");
        }
        =}
}
```



This instantiates the above Overwriting reactor and monitors its outputs.

Method Declaration

Causality Loops

Cycles

The interconnection pattern for a collection of reactors can form a cycle, but some care is required. Consider the following example:

```
target C;
reactor A {
    input x:int;
    output y:int;
    reaction(x) -> y {=
        // ... something here ...
    =}
}
reactor B {
    input x:int;
    output y:int;
    reaction(x) {=
        // ... something here ...
    =}
    reaction(startup) -> y {=
        // ... something here ...
    =}
main reactor {
    a = new A();
    b = new B();
    a.y -> b.x;
    b.y \rightarrow a.x;
}
```

This program yields the following diagram:



The diagram highlights a **causality loop** in the program. At each tag, in reactor B, the first reaction has to execute before the second if it is enabled, a precedence indicated with the red dashed arrow. But the first can't execute until the reaction of A has executed, and that reaction cannot execute until the second reaction B has executed. There is no way to satisfy these requirements, so the tools refuse to generated code.

Cycles with Delays

One way to break the causality loop and get an executable program is to introduce a <u>logical delay</u> into the loop, as shown below:

```
target C;
reactor A {
    input x:int;
    output y:int;
    reaction(x) -> y {=
        // ... something here ...
    =}
}
reactor B {
    input x:int;
    output y:int;
    reaction(x) {=
        // ... something here ...
    =}
    reaction(startup) -> y {=
        // ... something here ...
    =}
}
main reactor {
    a = new A();
    b = new B();
    a.y -> b.x after 0;
    b.y -> a.x;
}
Lingua Franca diagram
```

Here, we have used a delay of 0, which results in a delay of one <u>microstep</u>. We could equally well have specified a positive time value.

Reaction Order

Frequently, a program will have such cycles, but you don't want a logical delay in the loop. To get a cycle without logical delays, the reactions need to be reordered, as shown below:

```
target C;
reactor A {
    input x:int;
    output y:int;
    reaction(x) -> y {=
        // ... something here ...
    =}
}
reactor B {
    input x:int;
    output y:int;
    reaction(startup) -> y {=
        // ... something here ...
    =}
    reaction(x) {=
        // ... something here ...
    =}
}
main reactor {
    a = new A();
    b = new B();
    a.y \rightarrow b.x;
    b.y \rightarrow a.x;
}
```

Lingua Franca diagram	

There is no longer any causality loop.

Extending Reactors

Extending a Base Reactor

Actions

Action Declaration

An action declaration has one of the following forms:

```
logical action <name>(<min_delay>, <min_spacing>, <policy>)
physical action <name>(<min_delay>, <min_spacing>, <policy>)
```

The min_delay, min_spacing, and policy are all optional. If only one argument is given in parentheses, then it is interpreted as an min_delay, if two are given, then they are interpreted as min_delay and min_spacing. The min_delay and min_spacing are time values. The policy argument is a string that can be one of the following: "defer" (the default), "drop", or "replace". Note that the quotation marks are needed.

Logical Actions

Timers are useful to trigger reactions once or periodically. Actions are used to trigger reactions more irregularly. An action, like an output or input port, can carry data, but unlike a port, an action is visible only within the reactor that defines it.

There are two kinds of actions, **logical** and **physical**. A **logical action** is used by a reactor to schedule a trigger at a fixed logical time interval d into the future. The time interval d, which is called a **delay**, is relative to the logical time t at which the scheduling occurs. If a reaction executes at logical time t and schedules an action a with delay d, then any reaction that is triggered by a will be invoked at logical time t + d. For example, the following reaction schedules something (printing the current elapsed logical time) 200 msec after an input x arrives:

```
target C;
reactor Schedule {
    input x:int;
    logical action a;
    reaction(x) -> a {=
        schedule(a, MSEC(200));
    =}
    reaction(a) {=
        interval_t elapsed_time = get_elapsed_logical_time();
        printf("Action triggered at logical time %lld nsec after start.\n",
    =}
}
```



Here, the delay is specified in the call to schedule() within the target language code. Notice that in the diagram, a logical action is shown as a triangle with an **L**. Logical actions are always scheduled within a reaction of the reactor that declares the action.

The arguments to the schedule() function are the action named a and a time. The action a has to be declared as an effect of the reaction in order to reference it in the call to schedule(). If you fail to declare it as an effect (after the -> in the reaction signature), then you will get an error message.

The time argument to the schedule() function is required to be non-negative. If it is zero, then the action will be scheduled one **microstep** later. See <u>Superdense Time</u> below.

The time argument to the <code>schedule()</code> function has data type <code>interval_t</code>, which, with the exception of some embedded platforms, is a C <code>long long</code>. A collection of convenience macros is provided like the <code>MSEC</code> macro above to specify time values in a more readable way. The provided macros are <code>NSEC</code>, <code>USEC</code> (for microseconds), <code>MSEC</code>, <code>SEC</code>, <code>MINUTE</code>, <code>HOUR</code>, <code>DAY</code>, and <code>WEEK</code>. You may also use the plural of any of these, e.g. <code>WEEKS(2)</code>.

An action may have a data type, in which case, a variant of the schedule() function can be used to specify a **payload**, a data value that is carried from where the schedule() function is called to

the reaction that is triggered by the action. See the <u>Target Language Reference</u>.

Physical Actions

A **physical action** is used to schedule reactions at logical times determined by the local physical clock. If a physical action with delay d is scheduled at *physical* time T, then the *logical time* assigned to the event is T + d. For example, the following reactor schedules the physical action p to trigger at a **logical time** equal to the **physical time** at which the input x arrives:

```
target C;
reactor Physical {
    input x:int;
    physical action a;
    reaction(x) -> a {=
        schedule(a, 0);
    =}
    reaction(a) {=
        interval_t elapsed_time = get_elapsed_logical_time();
        printf("Action triggered at logical time %lld nsec after start.\n",
        =}
}
```



If you drive this with a timer, using for example the following structure:



then running the program will yield an output something like this:

```
Action triggered at logical time 201491000 nsec after start. Action triggered at logical time 403685000 nsec after start. Action triggered at logical time 603669000 nsec after start.
```

Here, logical time is lagging physical time by a few milliseconds. Note that, unless the <u>fast option</u> is given, logical time t chases physical time T, so t < T. Hence, the event being scheduled in the reaction to input x is assured of being in the future in logical time.

Whereas logical actions are required to be scheduled within a reaction of the reactor that declares the action, physical actions can be scheduled by code that is outside the Lingua Franca system. For example, some other thread or a callback function may call schedule(), passing it a physical action. For example:

```
target C;
main reactor {
        preamble {=
                // Schedule an event roughly every 200 msec.
                void* external(void* a) {
            while (true) {
                        lf nanosleep(MSEC(200));
                        schedule(a, 0);
                        }
                }
        =}
        state thread_id:lf_thread_t(0);
    physical action a(100 msec):int;
        reaction(startup) -> a {=
                // Start a thread to schedule physical actions.
                lf_thread_create(&self->thread_id, &external, a);
        =}
        reaction(a) {=
        interval t elapsed time = get elapsed logical time();
        printf("Action triggered at logical time %lld nsec after start.\n",
        =}
}
Lingua Franca diagram
```



Physical actions are the mechanism for obtaining input from the outside world. Because they are assigned a logical time derived from the physical clock, their logical time can be interpreted as a measure of the time at which some external event occurred.

In the above example, at **startup**, the main reactor creates an external thread that schedules a physical action roughly every 200 msec. The thread uses a built-in function <code>lf_nanosleep()</code>, which abstracts platform-specific mechanisms for stalling the thread for a specified amount of time. The thread is created with a built-in function <code>lf_thread_create()</code>, which similarly abstracts platform-specific mechanisms for creating threads.

The code executed by the thread is defined in a **preamble** section. See <u>Preambles and Methods</u>.

Important Note: Asynchronous calls to schedule() will not work if you set the threading target parameter to false. You must use a threaded runtime for such asynchronous calls to work correctly.

Triggering Time for Actions

An action will trigger at a logical time that depends on the arguments given to the schedule function, the <min_delay> , <min_spacing> , and <policy> arguments in the action declaration, and whether the action is physical or logical.

For a **logical** action a, the tag assigned to the event resulting from a call to schedule(a, <offset>) is computed as follows. First, let t be the current logical time. For a logical action, t is just the logical time at which the reaction calling schedule() is called. The **preliminary time** of the action is then just $t + \langle \min_{delay} + \langle offset \rangle$. This preliminary time may be further modified, as explained below.

For a **physical** action, the preliminary time is similar, except that *t* is replaced by the current *physical* time *T* when schedule() is called.

If a <min_spacing> has been declared, then it gives a minimum logical time interval between the tags of two subsequently scheduled events. If the preliminary time is closer than <min_spacing> to the time of the previously scheduled event (if there is one), then <policy> determines how the minimum spacing constraint is enforced. The <policy> is one of the following:

- "defer": (**the default**) The event is added to the event queue with a tag that is equal to earliest time that satisfies the minimal spacing requirement. Assuming the time of the preceding event is *t_prev*, then the tag of the new event simply becomes *t_prev* + <min spacing>.
- "drop": The new event is dropped and schedule() returns without having modified the event queue.

• "replace": The payload (if any) of the new event is assigned to the preceding event if it is still pending in the event queue; no new event is added to the event queue in this case. If the preceding event has already been pulled from the event queue, the default "defer" policy is applied.

Note that while the "defer" policy is conservative in the sense that it does not discard events, it could potentially cause an unbounded growth of the event queue.

Superdense Time

Tag vs. Time

The model of time in Lingua Franca is a bit more sophisticated than we have hinted at. Specifically, a **superdense** model of time is used. In particular, instead of a **timestamp**, LF uses a **tag**, which consists of a **logical time** *t* and a **microstep** *m*.

A <u>logical action</u> may have a <u>schedule()</u> function may be zero. In this case, the call to <u>schedule()</u> appears to be requesting that the action trigger at the *current logical time*. Here is where superdense time comes in. The action will indeed trigger at the current logical time, but one microstep later. Consider the following example:

```
target C;
main reactor {
    state count:int(1);
    logical action a;
    reaction(startup, a) -> a {=
        printf("%d. Logical time is %lld. Microstep is %d.\n",
            self->count, get_logical_time(), get_microstep()
        );
        if (self->count++ < 5) {
            schedule(a, 0);
        }
        =}
}</pre>
```



Executing this program will yield something like this:

```
    Logical time is 1649607749415269000. Microstep is 0.
    Logical time is 1649607749415269000. Microstep is 1.
    Logical time is 1649607749415269000. Microstep is 2.
    Logical time is 1649607749415269000. Microstep is 3.
    Logical time is 1649607749415269000. Microstep is 4.
```

Notice that the logical time is not advancing, but the microstep is (the logical time, in this case, gives the number of nanoseconds that have elapsed since January 1, 1970). The general rule is that **every** call to schedule() advances the tag by at least one microstep.

Logical Simultaneity

Two events are **logically simultaneous** only if *both* the logical time and the microstep are equal. The following example illustrates this:

```
target C;
reactor Destination {
    input x:int;
    input y:int;
    reaction(x, y) {=
        printf("Time since start: %lld, microstep: %d\n",
            get elapsed logical time(), get microstep()
        );
        if (x->is_present) {
            printf(" x is present.\n");
        }
        if (y->is_present) {
            printf(" y is present.\n");
        }
    =}
}
main reactor {
    logical action repeat;
    d = new Destination();
    reaction(startup) -> d.x, repeat {=
        SET(d.x, 1);
        schedule(repeat, 0);
    =}
    reaction(repeat) -> d.y {=
        SET(d.y, 1);
    =}
}
```



The Destination reactor has two inputs, x and y, and it reports in a reaction to either input what is the logical time, the microstep, and which input is present. The main reactor reacts to startup by sending data to the x input of Destination. It then schedules a repeat action with an <offset> of zero. The repeat reaction is invoked strictly later, one microstep later. The output printed, therefore, will look like this:

```
Time since start: 0, microstep: 0
  x is present.
Time since start: 0, microstep: 1
  y is present.
```

The time reported by <code>get_elapsed_logical_time()</code> has not advanced in the second reaction, but the fact that <code>x</code> is not present in the second reaction proves that the first reaction and the second are not logically simultaneous. The second occurs one microstep later.

Alignment of Logical and Physical Times

Recall that in Lingua Franca, logical time "chases" physical time, invoking reactions at a physical time close to their logical time. For that purpose, the microstep is ignored.

Deadlines

Lingua Franca includes a notion of a **deadline**, which is a constraint on the relation between logical time and physical time. Specifically, a program may specify that the invocation of a reaction must occur within some *physical* time interval of the *logical* time of the message. If a reaction is invoked at logical time 12 noon, for example, and the reaction has a deadline of one hour, then the reaction is required to be invoked before the physical-time clock of the execution platform reaches 1 PM. If the deadline is violated, then the specified deadline handler is invoked instead of the reaction. For example:

```
target C;
reactor Deadline {
    input x:int;
    output d:int; // Produced if the deadline is violated.
    reaction(x) -> d {=
        printf("Normal reaction.\n");
    =} deadline(10 msec) {=
        printf("Deadline violation detected.\n");
        SET(d, x->value);
    =}
}
```

This reactor specifies a deadline of 10 milliseconds (this can be a parameter of the reactor). If the reaction to x is triggered later in physical time than 10 msec past the timestamp of x, then the second body of code is executed instead of the first. That second body of code has access to anything the first body of code has access to, including the input x and the output x and the output can be used to notify the rest of the system that a deadline violation occurred. This reactor can be tested as follows:

```
target C;
import Deadline from "Deadline.lf";
main reactor {
    logical action a;
    d = new Deadline();
    reaction(startup) -> d.x, a {=
        SET(d.x, 0);
        schedule(a, 0);
    =}
    reaction(a) -> d.x {=
        SET(d.x, 0);
        lf_nanosleep(MSEC(20));
    =}
    reaction(d.d) {=
        printf("Deadline reactor produced an output.\n");
    =}
}
```

ELingua Franca diagram

Running this program will result in the following output:

Normal reaction.

Deadline violation detected.

Deadline reactor produced an output.

The first reaction of the <code>Deadline</code> reactor does not violate the deadline, but the second does. Notice that the sleep in the <code>main</code> reactor occurs <code>after</code> setting the output, but because of the deterministic semantics of LF, this does not matter. The actual value of an output cannot be known until every reaction that sets that output <code>completes</code> its execution. Since this reaction takes at least 20 msec to complete, the deadline is assured of being violated.

Notice that the deadline is annotated in the diagram with a small clock symbol.

Multiports and Banks

Lingua Franca provides a compact syntax for ports that can send or receive over multiple channels and another syntax for multiple instances of a reactor class. These are respectively called **multiports** and **banks of reactors**.

Multiports

To declare an input or output port to be a **multiport**, use the following syntax:

where <width> is a positive integer. This can be given either as an integer literal or a parameter name. Consider the following example:

```
target C;
reactor Source {
    output[4] out:int;
    reaction(startup) -> out {=
        for(int i = 0; i < out width; i++) {
            SET(out[i], i);
        }
    =}
}
reactor Destination {
    input[4] in:int;
    reaction(in) {=
        int sum = 0;
        for (int i = 0; i < in width; i++) {
            if (in[i]->is present) sum += in[i]->value;
        printf("Sum of received: %d.\n", sum);
    =}
}
main reactor {
    a = new Source();
    b = new Destination();
    a.out -> b.in;
}
```



Executing this program will yield:

Sum of received: 6.

The Source reactor has a four-way multiport output and the Destination reactor has a four-way multiport input. These channels are connected all at once on one line, the second line from the last. Notice that the generated diagram shows multiports with hollow triangles. Whether it shows the widths is controlled by an option in the diagram generator.

NOTE: In Destination, the reaction is triggered by in, not by some individual channel of the multiport input. Hence, it is important when using multiport inputs to test for presence of the input on each channel, as done above with the syntax if (in[i]->is_present) An event on any one of the channels is sufficient to trigger the reaction.

The Source reactor also specifies out as an effect of its reaction using the syntax -> out. This brings into scope of the reaction body a way to access the width of the port and a way to write to each channel of the port.

Parameterized Widths

The width of a port may be given by a parameter. For example, the above Source reactor can be rewritten

```
reactor Source(width:int(4)) {
   output[width] out:int;
   reaction(startup) -> out {=
        ...
   =}
}
```

Connecting Reactors with Different Widths

Assume that the Source and Destination reactors above both use a parameter width to specify the width of their ports. Then the following connection is valid:

```
main reactor {
    a1 = new Source(width = 3);
    a2 = new Source(width = 2);
    b = new Destination(width = 5);
    a1.out, a2.out -> b.in;
}
```

The first three ports of b will received input from a1, and the last two ports will receive input from a2. Parallel composition can appear on either side of a connection. For example:

```
al.out, a2.out -> b1.out, b2.out, b3.out;
```

If the total width on the left does not match the total width on the right, then a warning is issued. If the left side is wider than the right, then output data will be discarded. If the right side is wider than the left, then inputs channels will be absent.

Any given port can appear only once on the right side of the -> connection operator, so all connections to a multiport destination must be made in one single connection statement.

Banks of Reactors

Using a similar notation, it is possible to create a bank of reactors. For example, we can create a bank of four instances of Source and four instances of Destination and connect them as follows:

```
main reactor {
    a = new[4] Source();
    b = new[4] Destination();
    a.out -> b.in;
}
```



If the Source and Destination reactors have multiport inputs and outputs, as in the examples above, then a warning will be issued if the total width on the left does not match the total width on the right. For example, the following is balanced:

```
main reactor {
    a = new[3] Source(width = 4);
    b = new[4] Destination(width = 3);
    a.out -> b.in;
}
```

There will be three instances of Source, each with an output of width four, and four instances of Destination, each with an input of width 3, for a total of 12 connections.

To distinguish the instances in a bank of reactors, the reactor can define a parameter called **bank_index**. If such a parameter is defined for the reactor, then when the reactor is instanced in a bank, each instance will be assigned a number between 0 and *n*-1, where *n* is the number of reactor instances in the bank. For example, the following source reactor increments the output it produces by the value of bank_index on each reaction to the timer:

```
target C;
reactor MultiportSource(
    bank index:int(0)
) {
    timer t(0, 200 \text{ msec});
    output out:int;
    state s:int(0);
    reaction(t) -> out {=
        SET(out, self->s);
        self->s += self->bank index;
    =}
}
The width of a bank may also be given by a parameter, as in
main reactor(
    source bank width:int(3),
    destination bank width:int(4)
) {
    a = new[source bank width] Source(width = 4);
    b = new[destination bank width] Destination(width = 3);
    a.out -> b.in;
}
```

Contained Banks

Banks of reactors can be nested. For example, note the following program:

```
target C;
reactor Child (
    bank index:int(0)
) {
    reaction(startup) {=
        printf("My bank index: %d.\n", self->bank index);
    =}
}
reactor Parent (
    bank index:int(0)
) {
    c = new[2] Child();
}
main reactor {
    p = new[2] Parent();
}
```



In this program, the Parent reactor contains a bank of Child reactor instances with a width of 2. In the main reactor, a bank of Parent reactors is instantiated with a width of 2, therefore, creating 4 Child instances in the program in total. The output of this program will be:

```
My bank index: 0.
My bank index: 1.
My bank index: 0.
My bank index: 1.
```

The order of these outputs will be nondeterministic if the execution is multithreaded (which it will be by default) because there is no dependence between the reactions, and, hence, they can execute in parallel.

The bank index of a container (parent) reactor can be passed down to contained (child) reactors. For example, note the following program:

```
target C;
reactor Child (
    bank index:int(0),
    parent bank index:int(0)
) {
    reaction(startup) {=
        printf(
            "My bank index: %d. My parent's bank index: %d.\n",
            self->bank index, self->parent bank index
        );
    =}
}
reactor Parent (
    bank_index:int(0)
) {
    c = new[2] Child(parent bank index = bank index);
}
main reactor {
    p = new[2] Parent();
}
```

In this example, the bank index of the Parent reactor is passed to the parent_bank_index parameter of the Child reactor instances. The output from this program will be:

```
My bank index: 1. My parent's bank index: 1. My bank index: 0. My parent's bank index: 0. My bank index: 0. My parent's bank index: 1. My bank index: 1. My parent's bank index: 0.
```

Again, note that the order of these outputs is nondeterministic.

Finally, members of contained banks of reactors can be individually addressed in the body of reactions of the parent reactor if their input/output port appears in the reaction signature. For example, note the following program:

```
target C;
reactor Child (
    bank index:int(0),
    parent bank index:int(0)
) {
    output out:int;
    reaction(startup) -> out {=
        SET(out, self->parent_bank_index * 2 + self->bank_index);
    =}
}
reactor Parent (
    bank index:int(0)
) {
    c = new[2] Child(parent bank index = bank index);
    reaction(c.out) {=
        for (int i=0; i < c_width; i++) {</pre>
            printf("Received %d from child %d.\n", c[i].out->value, i);
        }
    =}
}
main reactor {
    p = new[2] Parent();
}
```

Lingua Franca diagram

Running this program will give something like the following:

```
Received 0 from child 0.
Received 1 from child 1.
Received 2 from child 0.
Received 3 from child 1.
```

Combining Banks and Multiports

Banks of reactors may be combined with multiports, as in the following example:

```
target C;
reactor Source {
    output[3] out:int;
    reaction(startup) -> out {=
        for(int i = 0; i < out width; i++) {
            SET(out[i], i);
        }
    =}
}
reactor Destination(
    bank index:int(0)
) {
    input in:int;
    reaction(in) {=
        printf("Destination %d received %d.\n", self->bank_index, in->value
    =}
}
main reactor MultiportToBank {
    a = new Source();
    b = new[3] Destination();
    a.out -> b.in;
}
```



The three outputs from the Source instance a will be sent, respectively, to each of three instances of Destination, b[0], b[1], and b[2]. The result of the program will be something like the following:

```
Destination 0 received 0.
Destination 1 received 1.
Destination 2 received 2.
```

Again, the order is nondeterministic in a multithreaded context.

The reactors in a bank may themselves have multiports. In all cases, the number of ports on the left of a connection must match the number on the right, unless the ones on the left are iterated, as explained next.

Broadcast Connections

Occasionally, you will want to have fewer ports on the left of a connection and have their outputs used repeatedly to broadcast to the ports on the right. In the following example, the outputs from an ordinary port are broadcast to the inputs of all instances of a bank of reactors:

```
reactor Source {
        output out:int;
        reaction(startup) -> out {=
                 ... write to out ...
        =}
}
reactor Destination {
        input in:int;
        reaction(in) {=
                 ... read from in ...
        =}
}
main reactor ThreadedThreaded(width:int(4)) {
        a = new Source();
        d = new[width] Destination();
        (a.out) + -> d.in;
}
```

The syntax (a.out)+ means "repeat the output port a.out one or more times as needed to supply all the input ports of d.in." The content inside the parentheses can be a comma-separated list of ports, the ports inside can be ordinary ports or multiports, and the reactors inside can be ordinary reactors or banks of reactors. In all cases, the number of ports inside the parentheses on the left must divide the number of ports on the right.

Interleaved Connections

Sometimes, we don't want to broadcast messages to all reactors, but need more fine-grained control as to which reactor within a bank receives a message. If we have separate source and destination reactors, this can be done by combining multiports and banks as was shown in Combining Banks and Multiports. Setting a value on the index n of the output multiport, will result in a message to the n-th reactor instance within the destination bank. However, this pattern gets slightly more complicated, if we want to exchange addressable messages between instances of the same bank. This pattern is shown in the following example:

```
target C;
reactor Node(
    num nodes: size t(4),
    bank index: int(0)
) {
    input[num nodes] in: int;
    output[num nodes] out: int;
    reaction (startup) -> out {=
        SET(out[1], 42);
        printf("Bank index %d sent 42 on channel 1.\n", self->bank index);
    =}
    reaction (in) {=
        for (int i = 0; i < in width; i++) {
            if (in[i]->is present) {
                printf("Bank index %d received %d on channel %d.\n",
                    self->bank index, in[i]->value, i
                );
            }
        }
    =}
}
main reactor(num nodes: size t(4)) {
    nodes = new[num nodes] Node(num nodes=num nodes);
    nodes.out -> interleaved(nodes.in);
}
```

Lingua Franca diagram

In the above program, four instance of Node are created, and, at startup, each instance sends 42 to its second (index 1) output channel. The result is that the second bank member (bank_index 1)

will receive the number 42 on each input channel of its multiport input. Running this program gives something like the following:

```
Bank index 0 sent 42 on channel 1.
Bank index 1 sent 42 on channel 1.
Bank index 2 sent 42 on channel 1.
Bank index 3 sent 42 on channel 1.
Bank index 1 received 42 on channel 0.
Bank index 1 received 42 on channel 1.
Bank index 1 received 42 on channel 2.
Bank index 1 received 42 on channel 3.
```

In bank index 1, the 0-th channel receives from bank_index 0, the 1-th channel from bank_index 1, etc. In effect, the choice of output channel specifies the destination reactor in the bank, and the input channel specifies the source reactor from which the input comes.

This style of connection is accomplished using the new keyword **interleaved** in the connection. Normally, a port reference such as **nodes**.out where **nodes** is a bank and out is a multiport, would list all the individual ports by first iterating over the banks and then, for each bank index, iterating over the ports. If we consider the tuple (b,p) to denote the index b within the bank and the index p within the multiport, then the following list is created: (0,0), (0,1), (0,2), (0,3), (1,0), (1,1), (1,2), (1,3), (2,0), (2,1), (2,2), (2,3), (3,0), (3,1), (3,2), (3,3). However, if we use **interleaved** (nodes.out) instead, the connection logic will iterate over the ports first and then the banks, creating the following list: (0,0), (1,0), (2,0), (3,0), (0,1), (1,1), (2,1), (3,1), (0,2), (1,2), (2,2), (3,2), (0,3), (1,3), (2,3), (3,3). By combining a normal port reference with a interleaved reference, we can construct a fully connected network. The figure below visualizes this how this pattern would look without banks or multiports:



If we were to use a normal connection nodes.out -> nodes.in; instead of the **interleaved** connection, then the following pattern would be created:

Lingua Franca diagram

Effectively, this connects each reactor instance to itself, which isn't very useful.

Preambles and Methods

Preamble

Reactions may contain arbitrary target-language code, but often it is convenient for that code to invoke external libraries or to share procedure definitions. For either purpose, a reactor may include a **preamble** section.

For example, the following reactor uses the common stdlib C library to convert a string to an integer:

```
main reactor {
    preamble {=
          #include <stdlib.h>
    =}
    timer t;
    reaction(t) {=
          char* s = "42";
          int i = atoi(s);
          printf("Converted string %s to int %d.\n", s, i);
          =}
}
```

This will print:

Converted string 42 to int 42.

By putting the #include in the **preamble**, the library becomes available in all reactions of this reactor. Oddly, it also becomes available in all subsequently defined reactors in the same file or in files that include this file.

You can also use the **preamble** to define functions that are shared across reactions and reactors, as in this example:

```
main reactor {
    preamble {=
        int add_42(int i) {
            return i + 42;
        }
    =}
    timer t;
    reaction(t) {=
        printf("42 plus 42 is %d.\n", add_42(42));
    =}
}
Not surprisingly, this will print:
42 plus 42 is 84.
```

A **preamble** can also be put outside the **reactor** definition. Currently, in the C target, it makes no difference whether it is put inside or outside.

Methods

Distributed Execution

! The information in this page is outdated, and will be updated soon.

NOTE: This describes a highly preliminary capability to map pieces of a Lingua Franca program to different machines. This capability is very much under development. This capability has been tested on MacOS and Linux, but not yet on Windows. Volunteer to do that and update these instructions?

A Lingua Franca program can be separated into separate programs called **federates** that communicate with one another. The federates will execute in separate processes and even on separate machines. They can be distributed across networks and can even be written in different target languages.

There is always one federate named **RTI**, for **runtime infrastructure** that coordinates startup and shutdown and may, if the implementation is centralized, mediate communication. The RTI needs to be compiled and installed separately on the system before any federation can execute. The instruction on how to do so can be found here.

Other than the RTI, if there are *n* federates, then the code generator will generate *n* separate programs with names of the form *Name_Federate*, where *Name* is the name of the top-level Lingua Franca file (without the .lf extension) and *Federate* is the name of the federate reactor. The code generator also produces a shell script that lauches all the federates and the RTI and a second shell script that distributes the generated code for the federates (not the RTI) to the specified machines and compiles the code on that machine.

Minimal Example

A minimal federated execution is specified by using the **federated** keyword instead of **main** for the main federate. An example is given in <u>example/C/Federated/HelloWorld/HelloWorld.lf</u>, which looks like this:

```
target C;
reactor MessageGenerator { ... }
reactor PrintMessage { ... }

federated reactor DistributedHelloWorld {
    source = new MessageGenerator();
    print = new PrintMessage();
    source.message -> print.message;
}
```

The **federated** keyword tells the code generator that the program is to be split into several distinct programs, one for each top level reactor. When you run the code generator on example/C/Federated/HelloWorld/HelloWorld.If, the following three programs will appear in the bin directory:

- HelloWorld
- HelloWorld_source
- HelloWorld_print

The root name, *HelloWorld*, is the name of the .lf file from which these are generated. The suffixes "_source" and "_print" come from the names of the top-level instances. There will always be one federate for each top-level reactor instance.

To run the program, you can simply run bin/HelloWorld, which is a bash script that launches the other three programs. Alternatively, you can manually execute the RTI and the federate programs by starting them on the command line in any order.

In addition, one more bash shell scripts may be generated:

HelloWorld_distribute.sh

This script is generated if any of the two federates, or the RTI are specified to be run on a remote machine (see below for how to do that). This script will copy the source files for the relevant program (but not the RTI) to the remote machine and compile them there. The RTI needs to be separately installed on the remote machine.

Coordinated Start

When the above programs execute, each federate registers with the RTI. When all expected federates have registered, the RTI broadcasts to the federates the logical time at which they should start execution. Hence, all federates start at the same logical time.

The starting logical time is determined as follows. When each federate starts executing, it sends its current physical time (drawn from its real-time clock) to the RTI. When the RTI has heard from all the federates, it chooses the largest of these physical times, adds a fixed offset (currently one second), and broadcasts the resulting time to each federate.

When a federate receives the starting time from the RTI, if it is running in realtime mode (the default), then it will wait until its local physical clock matches or exceeds that starting time. Thus, to the extent that the machines have <u>synchronized clocks</u>, the federates will all start executing at roughly the same physical time, a physical time close to the starting logical time.

Coordinated Shutdown

Coordinating the shutdown of a distributed program is discussed in [[Termination]].

Communication Between Federates

When one federate sends data to another, by default, the timestamp at the receiver will match the timestamp at the sender. You can also specify a logical delay on the communication using the **after** keyword. For example, if we had instead specified

source.out -> print.in after 200 msec;

then the timestamp at the receiving end will be incremented by 200 msec compared to the timestamp at the sender (see example/c/Federated/HelloWorld/HelloWorldAfter.lf).

The preservation of timestamps across federates implies some constraints (see physical.connections below for a way to avoid these constraints). How these constraints are managed depends on whether you choose **centralized** or **decentralized** coordination.

Containerized Execution

FIXME: Point to /docs/handbook/containerized-execution

FIXME: <u>Here</u> is an test that imports an <u>existing</u> federated test with the addition of a docker: true flag in the target property of the test. This test will automatically run in multiple Docker containers (one for the RTI and one for each federate) in our CI.

Centralized Coordination

In the **centralized** mode of coordination (the default), the RTI regulates the advancement of time in each of the federates in order to ensure that the logical time semantics of Lingua Franca is respected. If the **print** federate has an event with timestamp t that it wants to react to (it is the earliest event in its event queue or it is a **physical action** that just triggered), then it needs to get the OK from the RTI to advance its logical time to t. The RTI grants this time advance only when it can assure that **print** has received all messages that it will ever receive with timestamps t or less.

First, note that, by default, logical time on each federate never advances ahead of physical time, as reported by its local physical clock. Consider the consequences for the above connection. Suppose the timestamp of the message sent by source is *t*. This message cannot be sent before the local clock at source reaches *t* and also cannot be sent before the RTI grants to source a time advance to *t*. Since source has no federates upstream of it, the RTI will always grant it such a time advance.

Suppose that the communication latency is L. That is, it takes L time units (in physical time) for a message to traverse the network. Then the print federate will not see the message from source before physical time t + L, where this physical time is measured by the physical clock on source 's host. If that clock differs from the clock on print 's host by E, then print will see the message at physical time t + E + L, as measured by its own clock. Let the value of the **after** specification (200 msec above) be E. Then the timestamp of the received message is E0, therefore, will depend on the relationship between E1 and E3 and E4 and E5. If, for example, E5 and then federate print will lag behind physical time by at least E6 and E7 and E8 are the message in the relationship between E8 and E9 and E9 and E9 are the message in the relationship between E9 and E9 and E1 and E3 are the message in the message in the relationship between E6 and E8 are the message in the m

Assume the RTI has granted a time advance to t to federate source. Hence, source is able to send a message with timestamp t. The RTI now cannot grant any time advance to print that is greater than or equal to t + a until the message has been delivered to print. In centralized coordination, all messages flow through the RTI, so the RTI will deliver the time advance grant (**TAG**) to print only after it has delivered the message.

If a > E + L, then the existence of this communication does not cause print 's logical time to lag behind physical time. This means that if a **physical action** appears at print, the RTI will be able to immediately grant a time advance to print to the timestamp of that physical action. However, if a < E + L, then the RTI will delay granting a time advance to print by at least E + L - a. Hence, E + L - a represents an additional latency in the processing of physical actions! This latency could present a problem for meeting deadlines. For this reason, if there are physical actions or deadlines at a federate that receives network messages, it is desirable to set **after** on the connection to that federate to be larger than any expected E + L. This way, there is no additional latency to processing physical actions at this federate and no additional risk of missing deadlines.

If, in addition, the physical clocks on the hosts are allowed to drift with respect to one another, then *E* can grow without bound, and hence the lag between logical time and physical time in processing events can grow without bound. This is mitigated either by hosts that themselves realize some clock synchronization algorithm, such as <u>NTP</u> or <u>PTP</u>, or by utilizing Lingua Franca's own built in <u>clock synchronization</u>. If the federates lack physical actions and deadlines, however, then unsynchronized clocks present no problem if you are using centralized coordination.

With centralized coordination, all messages (except those on <u>physical connections</u>) go through the RTI. This can create a bottleneck and a single point of failure. To avoid this bottleneck, you can use decentralized coordination.

Decentralized Coordination

The default coordination between mechanisms is **centralized**, equivalent to specifying the target property:

coordination: centralized

Centralized coordination works as described above, where the advancement of time at each federate is regulated by the RTI. In order for the RTI to be able to safely grant a time advance to a federate, it is also necessary for all messages to that federate to go through the RTI. The RTI, therefore, can easily become a bottleneck.

An alternative is **decentralized** coordination, which uses a technique called <u>PTIDES</u>:

```
coordination: decentralized
```

This technique has also been implemented in Google Spanner, a globally distributed database system. In decentralized coordination, each federate has a **safe-to-process** (**STP**) offset. In decentralized coordination, when one federate communicates with another, it does so directly through a dedicated socket without going through the RTI. Moreover, it does not consult the RTI to advance logical time. Instead, it can advance its logical time to t when its physical clock matches or exceeds t + STP.

By default, the STP is zero. This will work fine under the assumption that **every** logical connection between federates has a sufficiently large **after** clause. That is, the value of the logical delay must exceed the sum of the <u>clock synchronization</u> error *E*, the network latency bound *L*, and the time lag on the sender *D* (the physical time at which it sends the message minus the timestamp of the message). The sender's time lag *D* can be enforced by using a **deadline**. See for example <u>example/C/Federated/HelloWorld/HelloWorld/Decentralized.lf</u>.

Of course, this assumption can be violated in practice. Analogous to a deadline violation, Lingua Franca provides a mechanism for handling such a violation that is called a tardy handlers as done in example/C/Federated/HelloWorld/HelloWorldDecentralized.lf.

(example/C/Federated/HelloWorld/HelloWorldDecentralized.lf). The pattern is:

```
reaction(in) {=
    // User code
=} tardy {=
    // Error handling code
=}
```

If the timestamp at which this reaction is to be invoked (the value returned by <code>get_current_tag</code>) cannot match the timestamp of an incoming message <code>in</code> (because the current tag has already advanced beyond the intended tag of <code>in</code>), then the <code>tardy</code> handler will be invoked instead of the normal reaction. Within the body of the tardy handler, the code can access the intended tag of <code>in</code> using <code>in->intended_tag</code>, which has two fields, a timestamp <code>in->intended_tag.time</code> and a microstep <code>in->intended_tag.microstep</code>. The code can then ascertain the severity of the error and act accordingly. If no tardy handler is provided at any reaction triggered by an input from

another federate, then the normal reaction will be invoked at the earliest feasible logical time greater than or equal to the intended logical time of the message.

One option available to the code is to increase the STP. This can be done simply by equipping a federate with a parameter of type **time** named STP. See for example example/C/Federated/HelloWorld/HelloWorldDecentralizedSTP.If. This can be done as follows:

```
import PrintMessageWithDetector from "HelloWorldDecentralized.lf"
reactor PrintMessageWithSTP(STP:time(10 msec)) extends PrintMessageWithDete
```

Notice that the only override in PrintMessageWithSTP is the addition of an STP parameter.

The LF API provides two functions that can be used to dynamically adjust the STP:

```
interval_t get_stp_offset();
void set_stp_offset(interval_t offset);
```

Using these functions, however, is a pretty advanced operation.

FIXME: The discussion of cycles in the remainder of this section needs to be revisited with pointers to newer examples.

Now suppose that if there are cycles in the communication between federates. For example, in addition to the above connection, suppose we also have a connection going in the opposite direction:

```
print.out -> count.in after 100 msec;
```

Now we potentially have a very big problem. The physical clock at print has to lag behind physical time by at least E + L - 200 msec, and the physical clock at count has to lag behind physical time by at least E + L - 100 msec. The latter of these means that count cannot send a message with timestamp t until its local clock exceeds t + E + L - 100 msec. If E + L - 100 msec > 0, then this additional lag increases the required lag at print, which will need to lag behind physical time now by E + L - 100 msec + E + L - 200 msec. If this number is positive, then the lag required at count will have to be again increased, which will then cause this number to again increase, and so on until the required lag is infinite at both federates. Thus, a cycle between two federates is **infeasible** if $2E + 2L - a_1 - a_2 > 0$, where a_1 and a_2 are the **after** values of the two connections. More generally, the sum of $E + L - a_1$ over all connections i in a cycle must be less than or equal to zero. Otherwise, decentralized coordination will fail and finite STP will lead to tardy messages. Centralized coordination can be used instead if the program really must be this way.

The bottom line is that if there are cycles in your federation and/or you have physical actions in federates that receive network messages, it is wise to specify **after** to be larger than the sum of the greatest expect clock synchronization error *E* and the greatest expected network latency *L*.

Physical Connections

Coordinating the execution of the federates so that timestamps are preserved is tricky. If your application does not require the deterministic execution that results from preserving the timestamps, then you can alternatively specify a **physical connection** as follows (see example/C/Federated/HelloWorld/HelloWorldPhysical.lf:

```
source.out ~> print.in;
```

The tilde specifies that the timestamp of the sender should be discarded. A new timestamp will be assigned at the receiving end based on the local physical clock, much like a **physical action**. To distinguish it from a physical connection, the normal connection is called a **logical connection**.

There are a number of subtleties with physical connections. One is that if you specify an after clause, for example like this:

```
count.out ~> print.in after 10 msec;
```

then what does this mean? At the receiving end, the timestamp assigned to the incoming event will be the current physical time plus 10 msec.

Prerequisites for Distributed Execution

In the above example, all of the generated programs expect to run on localhost. This is the default. With these defaults, every federate has to run on the same machine as the RTI because localhost is not a host that is visible from other machines on the network. In order to run federates or the RTI on remote machines, you can specify a domain name or IP address for the RTI and/or federates.

In order for a federated execution to work, there is some setup required on the machines to be used. First, each machine must be running on ssh server. On a Linux machine, this is typically done with a command like this:

```
sudo systemctl <start|enable> ssh.service
```

Enable means to always start the service at startup, whereas start means to just start it this once. On MacOS, open System Preferences from the Apple menu and click on the "Sharing" preference panel. Select the checkbox next to "Remote Login" to enable it. **FIXME**: Windows?

It will also be much more convenient if the launcher does not have to enter passwords to gain access to the remote machine. This can be accomplished by installing your public key (typically found in ~/.ssh/id_rsa.pub) in ~/.ssh/authorized_keys on the remote host.

Second, the RTI must be installed on the remote machine. Instructions about installation of RTI can be found here.

Specifying RTI Hosts

You can specify a domain name on which the RTI should run as follows:

```
federated reactor DistributedCount at www.example.com {
    ...
}
```

You can alternatively specify an IP address (either IPv4 or IPv6):

```
federated reactor DistributedCount at 10.0.0.198 { ... }
```

By default, the RTI starts a socket server on port 15045, if that port is available, and increments the port number by 1 until it finds an available port. The number of increments is limited by a target-specific number. In the C target, in rti.h, STARTING_PORT defines the number 15045 and PORT_RANGE_LIMIT limits the range of ports attempted (currently 1024).

You can also specify a port for the RTI to use as follows:

```
federated reactor DistributedCount at 10.0.0.198:8080 { ... }
```

If you specify a specific port, then it will use that port if it is available and fail otherwise. The above changes this to port 8080.

You can also specify a user name on the remote machine for cases where the username will not match whoever launches the federation:

```
federated reactor DistributedCount at user@10.0.0.198:8080 { ... }
```

The general form of the host designation is

```
federated reactor DistributedCount at user@host:port/path { ... }
```

where user@, :port, and /path are all optional. The path specifies the directory on the remote machine (relative to the home directory of the user) where the generated code will be put. The host should be an IPv4 address (e.g. 93.184.216.34), IPv6 address (e.g. 2606:2800:220:1:248:1893:25c8:1946), or a domain name (e.g. www.example.com). It can also be localhost or 0.0.0.0. The host can be remote as long as it is accessible from the machine where the programs will be started.

If user@ is not given, then it is assumed that the username on the remote host is the same as on the machine that launches the programs. If :port is not given, then it defaults to port 15045. If /path is not given, then ~user/LinguaFrancaRemote will be the root directory on the remote machine.

FIXME: Not implemented yet: If the IP address or hostname does not match the local machine on which code generation is being done, ...

A Federation_distribute.sh shell script will be generated. This script will distribute the generated code for the RTI to the remote machine at the specified directory.

Specifying Federate Hosts

A federate may be mapped to a particular remote machine using a syntax like this:

```
count = new Count() at user@host:port/path;
```

The port is ignored in **centralized** mode because all communication is routed through the RTI, but in **decentralized** mode it will specify the port on which a socket server listens for incomming connections from other federates.

If any federate (or the RTI) has such a remote designator, then a Federation_distribute.sh shell script will be generated. This script will distribute the generated code for the RTI to the remote machine at the specified directory.

Note that if the machine uses DHCP to obtain its address, then the generated code may not work in the future since the address of the machine may change in the future.

Address 0.0.0.0: In the above example, localhost is used. This is the default if no address is specified. Using localhost specifies that the generated programs should establish connections only with processes running on the local machine. This is ideal for testing. If you use 0.0.0.0, then you are also specifying that the local machine (the one performing the code generation) will be the host, but now the process(es) running on this local machine can establish connections with processes on remote machines. The code generator will determine the IP address of the local machine, and any other hosts that need to communicate with reactors on the local host will use the current IP address of that local host at the time of code generation.

Clock Synchronization

Both centralized and decentralized coordination have some reliance on clock synchronization. First, the RTI determines the start time of all federates, and the actually physical start time will differ by the extent that their physical clocks differ. This is particularly problematic if clocks differ by hours or more, which is certainly possible. If the hosts on which you are running run a clock synchronization algorithm, such as NTP or PTP, then you may not need to be concerned about this at all. Windows, Mac, and most versions of Linux, by default, run NTP, which synchronizes their clocks to some remote host. NTP is not particularly precise, however, so clock synchronization error can be hundreds of milliseconds or larger. PTP protocols are much more precise, so if your hosts derive

their physical clocks from a PTP implementation, then you probably don't need to do anything further. Unfortunately, as of this writing, even though almost all networking hardware provides support for PTP, few operating systems utilize it. We expect this to change when people have finally understood the value of precise clock synchronization.

If your host is not running any clock synchronization, or if it is running only NTP and your application needs tighter latencies, then Lingua Franca's own built-in clock synchronization may provide better precision, depending on your network conditions. Like NTP, it realizes a software-only protocol, which are much less precise than hardware-supported protocols such as PTP, but if your hosts are on the same local area network, then network conditions may be such that the performance of LF clock synchronization will be much better than NTP. If your network is equipped with PTP, you will want to disable the clock synchronization in Lingua Franca by specifying in your target properties the following:

clock-sync: off

When a federation is mapped onto multiple machines, then, by default, any federate mapped to a machine that is not the one running the RTI will attempt during startup to synchronize its clock with the one on the machine running the RTI. The determination of whether the federate is running on the same machine is determined by comparing the string that comes after the at clause between the federate and the RTI. If they differ at all, then they will be treated as if the federate is running on a different machine even if it is actually running on the same machine. This default behavior can be obtained by either specifying nothing in the target properties or saying:

clock-sync: initial

This results in clock synchronization being done during startup only. To account for the possibility of your clocks drifting during execution of the program, you can alternatively specify:

clock-sync: on

With this specification, in addition to synchronization during startup, synchronization will be redone periodically during program execution.

Clock Synchronization Options

A number of options can be specified using the clock-sync-options target parameter. For example:

clock-sync-options: {local-federates-on: true, test-offset: 200 msec}

The supported options are:

• local-federates-on: Should be true or false. By default, if a federate is mapped to the same host as the RTI (using the at keyword), then clock synchronization is turned off. This

assumes that the federate will be using the same clock as the RTI, so there is no point in performing clock synchronization. However, sometimes it is useful to force clock synchronization to be run even in this case, for example to test the performance of clock synchronization. To force clock synchronization on in this case, set this option to true.

- test-offset: The value should be a time value with units, e.g. 200 msec. This will establish an artificial fixed offset for each federate's clock of one plus the federate ID times the time value given. For example, with the value 200 msec, a fixed offset of 200 milliseconds will be set on the clock for federate 0, 400 msec on the clock of federate 1, etc.
- period: A time value (with units) that specifies how often runtime clock synchronization will be performed if it is turned on. The default is 5 msec.
- attenuation: A positive integer specifying a divisor applied to the estimated clock error during runtime clock synchronization when adjusting the clock offset. The default is 10. Making this number bigger reduces each adjustment to the clock. Making the number equal to 1 means that each round of clock synchronization fully applies its estimated clock synchronization error.
- trials: The number of rounds of message exchange with the RTI in each clock synchronization round. This defaults to 10.

Future Work

The RTI can also play the role of **auth**, an authentication and authorization server that ensures that only the generated programs can establish connections with each other and that their communication is encrypted, as explained in the <u>Security</u> section below.

Currently, the threads option is same on all federates. We need a mechanism to customize this parameter by federate.

Security

In addition to generating a program for each host, the code generator could generate configuration files for a program called **auth** designed to run on the first host that is preconfigured to provide authentication and authorization to each of the other generated programs together with encryption keys that are used for communicating between them. The auth program should be started first since non of the other generate programs will be able to authenticate without it.

The auth program, written by Hokeun Kim, comes from https://github.com/iotauth/iotauth and provides "locally centralized, globally distributed" authentication and authorization. Papers describing this work can be found here: [IoTDI '17], [FiCloud '16] [IT Professional '17'].

Protobufs

Communication between hosts can only be accomplished on channels where the message types are either language primitives or <u>Protobufs</u>. All other datatypes will be reject at code generation time.

Termination

Shutdown Reactions

There are several mechanisms for terminating a Lingua Franca in an orderly fashion. All of these mechanisms result in a **final tag** at which any reaction that declares **shutdown** as a trigger will be invoked (recall that a **tag** is a tuple (**logical time**, **microstep**)). Other reactions may also be invoked at this final tag, and the order in which reactions are invoked will be constrained by the normal precedence rules.

If a reaction triggered by **shutdown** produces outputs, then downstream reactors will also be invoked at the final tag. If the reaction schedules any actions by calling **schedule()**, those will be ignored. In fact, any event after the final tag will be ignored. After the completion of the final tag, the program will exit.

There are four ways to terminate a program:

- **Timeout**: The program specifies the last logical time at which reactions should be triggered.
- **Starvation**: At the conclusion of some tag, there are no events in the event queue at future tags.
- **Stop request**: Some reaction requests that the program terminate.
- **External signal**: Program is terminated externally using operating services like control-C or kill.

We address each of these in turn.

Timeout

The <u>target property timeout</u> specifies the last logical time at which reactions should be triggered. The last invocation of reactions will be at tag (timeout, 0).

There is a significant subtlety when using <u>physical connections</u>, which are connections using the syntax ~> . Such connections specify that the tag at the receiving end will be based on the physical time at which the message is received. If the tag assigned at the receiving end is greater than the final tag, then the message is lost. Hence, **messages sent near the timeout time are likely to be lost!**

Starvation

If a Lingua Franca program has no <u>physical actions</u>, and if at any time during execution there are no future events waiting to be processed, then there is no possibility for any more reactions to occur and the program will exit. This situation is called **starvation**. If there is a **timer** anywhere in the program with a period, then this condition never occurs.

One subtlety is that reactions triggered by **shutdown** will be invoked one microstep later than the last tag at which there was an event. They cannot be invoked at the same tag because it is only after that last tag has completed that the runtime system can be sure that there are no future events. It would not be correct to trigger the **shutdown** reactions at that point because it would be impossible to respect the required reaction ordering.

Stop Request

If a reaction calls the built-in <code>request_stop()</code> function, then it is requesting that the program cease execution as soon as possible. This cessation will normally occur in the next microstep. The current tag will be completed as normal. Then the tag will be advanced by one microstep, and reactions triggered by <code>shutdown</code> will be executed, along with any other reactions with triggers at that tag, with all reactions executed in precedence order.

External Signal

A control-C or other kill signal to a running Lingua Franca program will cause execution to stop immediately.